
//***************************************************************
//* MACRO: DCConvert.WCM
//* PURPOSE: Converts "first whole word" drop caps to "number of characters" drop caps 
for conversion to Word.
//***************************************************************

Application(A1; "WordPerfect"; Default!; "EN")

RevealCodesWasActivated = FALSE
If( ?RevealCodesActive() = FALSE )

// Ensure that Reveal Codes is on for character count accuracy
RevealCodes(On!)
RevealCodesWasActivated = TRUE

Endif

//Stop execution if document is empty
If( ?DocBlank)

Go( Quit@)
EndIf

PosDocVeryTop ()

While( 1)
// Search for the next drop capped word.
OnNotFound ( Quit@ )
SearchString (StrgToLookFor: "[Dropcap Definition]")
SearchNext (SearchMode: Extended!)

// Convert it into drop cap characters
Call (Convert)

EndWhile

Label( Quit@ )
If( RevealCodesWasActivated )

// Turn reveal codes back off if WE turned it on for this script
RevealCodes(Off!)

Endif

//******************************************************************
//Function Name: Convert
//Purpose: Converts whole word drops caps to number of character drop caps
//******************************************************************
Procedure Convert()



DropCapBegin()
vChars = ?DropCapCharacters
DropCapEnd(Cancel!)

If (vChars = 0) //Is Drop Caps set to first whole word?
vCount = 0

PosWordNext () //Position cursor between words to prepare for char count
While ( (?LeftChar = "") OR (?LeftChar = " ")  )

PosCharPrevious()
EndWhile

//Count the number of letters in the dropcapped word
While( ?LeftCode != 6099)

PosCharPrevious()
If ( ?RightCode = 0  )

vCount = vCount + 1
EndIf 

EndWhile

PosLineBeg ()

//Set the drop cap to the number of letters in the word
DropCapBegin ()
DropCapCharacters (Number: vCount)
DropCapEnd (State: Save!)

EndIf
EndProc
//*************************************************************************


